
Итак, мой пц починен методом перепаивания двух конденсаоров, и я

снова с вами.

У  нас на завод кто-то пронёс вирус, в связи с этим был объявлен

шмон; в частности, закрыли доступ в интернет (я, впрочем, не был

туда записан).

Даже  рассказик  придумался  по  этому  поводу, в стиле древнего

фидошного юмора:

————————————————————————————————————————————————————————————————

Касперский: Эй, задача, ты хто?

DrWeb: Я DrWeb, типа антивирус.

К: Это я антивирус! Я ставлюсь так, что меня фиг снесёшь!

D: А я нахожу вирусы, которые ещё не придуманы!

К: Зато я... замедляю файловые операции в два раза!

D: А я - перезагрузку в полтора!

К: А я... а я... а я тебя сейчас вылечу, выскочка несчастный!

D: Обломись, я тебя заблокировал.

Norton AntiVirus: Я вас щас обоих сотру.

К: А енто хто?

D: Нутром чую - вирус! Заблокирую-ка я его. Ну-ка, ну-ка...

N: В памяти обнаружен вирус DrWeb.

Windows: Закрываю DrWeb. Он блокировал Касперского. Разблокирую.

N: Товарищ юзер... я к тебе по вопросу о вирусе DrWeb.

   Стереть файл / Отформатировать диск?

Юзер: Отмена!!!

N: Как так отмена?.. Отмену мы не тестировали...

W: Выполнена недопустимая операция. Снимаю Norton.

   Перезагружаюсь...

Nero: Стой, я болванку не дорезал!

W: Nero не отвечает. Снять задачу?

Ю: Подождать!!!

К: Ага, у меня теперя 20 секунд. Ща я вылечу этот DrWeb, не

   будет выпендриваться :)

N: Эй ты, вирус! Думал, я одна задача и меня просто так сняли?

   Закрываю тебя нафиг и сотру.

К: Ручки твои загребущие. Я свою память защитил.

W: Выполнена недопустимая операция. Снимаю Norton.

   Перезагружаюсь...

N: Хрен ты перезагрузишься, пока я этого гада не стёр.

К: Ну-ну, так ты всю жизнь меня стирать будешь, а я пока твои

   запускальники поищу...

CPU: Я кипю.

Дрова: Температура процессора 100 градусов. Выключай комп!

W: Выключаю комп.

N: Ничего ты не выключишь, пока я не снимусь!

Ю: Чё-то всё повисло. Reset!

BIOS: Тестю память.

Память: Тестюсь.

CPU: БАБАХ!

————————————————————————————————————————————————————————————————

С  2  по  8  января  я был в Москве у Shiru Otaku. Мы одолжили у

Макса Тимонина  АТМку  и сделали  на ней игру "Time Gal" (игра с

древнего  игрового  автомата,  представляет  из  себя в основном

видео).  Конечно, сама игра делалась не всю неделю, ведь сначала

надо   было   изучить,  можно  ли  её  сделать  вообще - и  как,

собственно,  её делать. Сперва мы выясняли, как перегонять файлы

с пц на АТМ. Пц Shiru не захотел писать на ZX дискеты. Заработал

единственный способ - через MS-DOS дискету, используя iS-DOS'ные

утилиты.  Собственно, на этом первый день и закончился ( Тимонин

приехал  довольно  поздно,  и  все  эксперименты заняли довольно

долгое  время,  не считая показа демок, которые Макс очень хотел

нам показать и показал :)).

Потом  надо  было  переделать  под  АТМ  16-цветный видеоплейер,

который  я  писал  дома под пентагон. Как и ожидалось, запустить

плейер  не  удалось до тех пор, пока я не переделал HDDoctor под

АТМ  (я  пытался  ещё  летом у Чунина, но не вышло). При этом мы

выяснили,  что  CD-ROM,  стоящий  на  АТМ Тимонина, умеет читать

только  заводские  диски и Очень Неохотно может читать Некоторые

CD-R.  RW-шники  он  не  читает  в  принципе.  Поэтому  пришлось

поставить CD-ROM Shiru, и тут обнаружилось... что с ним на одном

шлейфе   не   дружит   винт.  Поэтому  мы  занялись  ди-джейским

перетыканием   CD-приводов.   Чтобы   проверить   каждую   новую

компиляцию программы, надо было:

1. Собрать загружаемую версию программы под эмулятором.

2. Урезать .trd с помощью SN.

3. Записать .trd на дискету.

4. Подключить к АТМ CD-привод Тимонина (иначе не работала запись

   на винт).

5. Включить АТМ.

6. Скопировать .trd с дискеты на винт.

7. Выключить АТМ.

8. Подключить к АТМ CD-привод Shiru (иначе CD-RW на читались).

9. Включить АТМ.

10. Запустить программу.

11. Выключить АТМ.

С горем пополам плейер удалось запустить. Потом синхронизировали

звук  (для  чего пришлось на слух определять скорость компьютера

:))).  Потом  убирали щелчки при переполнении буфера звука - для

этого  пришлось  положить в видеопоток больше звуковых данных и,

соответственно, меньше графических (понадобилось предусмотреть 1

сектор звука на каждые 7 секторов графики).

На  третий  день  Shiru  предложил идею: писать код игры на саму

болванку,  а  загрузчик с CD сделать раз и навсегда (списать его

один раз - и тогда больше ди-джействовать не придётся). Конечно,

чтобы отладить этот загрузчик, таки пришлось подиджействовать на

всю  катушку  :)  Вытаскивание файла игры из trdшника (ради чего

раньше  приходилось  запускать  SN ) Shiru смог автоматизировать

небольшой утилиткой.

Параллельно  я  создал  учётную  запись  на  zx.pk.ru  и пытался

проконсультироваться  там  хотя  бы  у кого-нибудь, но получил в

ответ  только  глупые  шутки.  Разве  что SMT смог уточнить, что

неэмулируемость   CD-операций  в  HDDoctor'е  (а  за  ним - и  в

видеоплейере)  происходит  из-за  кривой поддержки ATAPI-команды

#00  в  драйвере  (эта  команда  должна снимать любую предыдущую

ATAPI-команду).  А  без  #00 поддержку CD не написать. Разве что

через ATA-команду #EC (хитрость, найденная Budder'ом ), однако с

#EC  можно очень конкретно обломаться на ряде приводов - ведь её

командоснимательные способности не документированы.

Тем   временем   из   инета  докачалось  Delphi,  и  дальше  шли

эксперименты   с  видеоконвертором,  для  которых  ему  пришлось

скормить  кучу  видеороликов самого разного вида. На "Симпсонах"

мы  обнаружили  накопление  рассинхронизации  видео относительно

звука.  На  ролике  из  игры "Soul Edge" - плохое звучание после

передискретизации  без  интерполяции  и  ещё  рассыпание быстрых

фрагментов  видео,  которое  эффективнее всего лечится снижением

площади  картинки  (например,  если  картинка  на  15%  меньше -

пропускается на 25-30% меньше кадров!). На какой-то песне - сбой

в  выгрузке  файла  конвертором.  На  полнометражном мультфильме

"Призрак в доспехах" (там очень, доложу вам, реалистичные тени и

блики,  как  живые!  ничего  подобного  в  рисованной анимации я

раньше   не  видел) - пропуск  кадров  на  почти  не  меняющихся

участках. Заодно конвертор удалось ускорить примерно в 5 раз.

В пятницу-субботу я наконец дописал скриптовой движок. В ночь на

воскресенье  мы  конвертили  уровни. Shiru находил, какие ролики

(основная   последовательность  и  сцены  гибели)  соответствуют

конкретным   уровням,   доставал   из   них   картинки  и  звук,

нормализовал  звук  и  составлял  таблицы действий с указаниями,

какое   когда.  Потом  за  пц  садился  я - конвертил  ролики  в

спектрумовский  формат, писал скрипты, записывал болванку, тащил

её  на  АТМ  и  садился  тестировать  уровни.  В это время за пц

садился  Shiru  и  конвертил  следующие  уровни.  И так до утра.

Причём  пришлось  потратить  час на выяснение структуры каталога

CDFS.  С  определённого  момента  дескрипторы  файлов  перестали

помещаться  в  один сектор - и начались глюки. Выяснилось, что в

Абзаце#25  ошибка:  длина  дескриптора =#0000  означает не конец

каталога, а переход к следующему сектору!

В  воскресенье утром я захотел добавить в игру ещё немного кода,

но не получилось. Собрали как есть, с тем я и уехал.

Дома  я  портировал  игру  под пентагон с 16-цветной примочкой и

добавил загрузочную заставку (точнее, нашёл глюк, из-за которого

она  не  выводилась). Вот, собственно, и всё. Shiru выложил игру

на  своём  хостинге.  Теперь Макс не может сказать, что на сайте

http://atmturbo.narod.ru есть весь софт для АТМ :) Наша игра как

минимум раза в 2 больше этого сайта :)))

Ещё  в  Москве  я  успел  зарелизить  кучу самого разного софта,

который  долгие  годы  копился  у  меня  дома. В том числе много

исходников  я  отослал  Crank'у,  но...  ни  ответа, ни привета.

Что-то  случилось  с  его  сайтом  или  с ним самим, или и то, и

другое...

А  уехал  я  из  Москвы, вывозя огромную кучу скачанных сайтов с

библиотеками  (например,  огромная  www.krotov.info и знаменитый

www.vostlit.info ) и энциклопедиями ( www.1911encyclopedia.org и

ещё   несколько   маленьких,   которые  кинул  мылом  Константин

Свиридов ). Shiru  даже  умудрился  скачать  всю  Википедию,  но

распаковать её .tar не удалось - хитрая кодировка русских имён.

Пока  меня  не  было, Евгений Серёгин с помощью KSA-7G подключил

себе CMOS-часы. Протестировал их, а потом снова отключил :)

Недавно,  когда  ко  мне  приходил  KSA-7G,  мы  случайно сожгли

12-вольтовый  трансформатор  для  паяльника  (произошло короткое

замыкание  в  вилке  этого  самого  паяльника). Так что теперь я

паять не могу :( А у меня была мысля...

Приходил  также  John,  мы  даже  попытались  написать  какую-то

музыку.  Basil  сейчас  в  Тамбове. Интересно, что они с John'ом

купили себе GamePad для пц. Типа чтобы веселее было играть...

В  мыле  появлялся  Nikphe.  Он  рассказал,  что  теперь живёт и

работает в Рязани, что Event теперь пишет музыку на пц и что оба

они  часто  сидят  в  интернете.  Когда  я  спросил, какие у них

последние релизы, Nikphe мгновенно испарился :-/

У  Ldir'а,  как обычно, Last выпустить не получилось (как он сам

говорит, "пропил").

Я  существенно расширил и исправил список композиторов с помощью

Shiru  и  Newart'а. Теперь  известно 1053 (!) AY/YM композитора.

Подавляющее   большинство - спектрумовские   :)   И  подавляющее

большинство - из  России  :)))  Под  Turbo-Sound,  кстати,  пока

никто,  кроме  Shiru,  не  пишет,  хотя болтовни и обещаний было

много. А между тем, у Чунина уже есть рабочий экземпляр Turbo-FM

(где  вместо  YM2149  стоят  YM2203,  в  которых кроме обычных 3

каналов имеются FM-каналы).

Коммодорщики,   оказывается,   вполне   нормально   смотрят   на

распространение   огромных  баз  своих  адресов...  А  меня  вот

некоторые  спектрумисты  бранят  всё  время...  Я  уже  не  могу

по-человечески    зарелизить    новую   версию   своего   списка

спектрумистов, который куда подробнее старого.

Elph   прислал   мне   конвертор   графики  от  Baze,  он  вышел

одновременно  с  IG#8,  где  я  про  него  писал.  Он называется

"bzither".  Палитры  подбирает  очень удачно. Но не умеет делать

Floyd-Steinberg  diffusion  :(  Baze говорит, что есть конвертор

"dithvide" (by Pavel Cimbal ), который это умеет.

Выпущено: HDDoctor 0.10, Time Gal, VideConv.

————————————————————————————————————————————————————————————————

Поскольку кто первый встал - того и тапки, то:

               ZX-СТАHДАРТ HА АВТОЗАПУСКАЕМЫЕ CD

                              v1.0

Hа автозапускаемом CD (формат CDFS / ISO 9660 / ECMA-119) в

корневой директории должен располагаться файл AUTORUN.ZX (за

заглавные буквы можете не беспокоиться, Nero капитализует имена

файлов автоматически) длиной не более #8000 байт. Он грузится по

адресу #6000 (значение порта #7ffd равно #10) и запускается по

адресу #6000 с выключенными прерываниями IM1, при этом текущий

CD-привод проициализирован и является текущим устройством, кроме

того, произведена попытка включить на нём низкую скорость.

Регистры при этом:

SP=#6000

IY=23610

A=#A0 (master) или #B0 (slave) [возможно, будет выкинуто, т.к.

 1. на ZX уже устоялась ситуация CD-ROM=slave; 2. для игр,

 скорее  всего, не потребуется переключаться на другие

 устройства]

B=тип компьютера: 0=PENTAGON, 1=ATM, 2=SCORPION, 3=PROFI,

 4=SPRINTER

C=тип контроллера IDE: 0=NEMO, 1=ATM, 2=SMUC, 3=PROFI,

 4=SPRINTER

D=язык: 0=ENGLISH, 1=РУССКИЙ

E=адрес COVOX: #FB=ATM/PENTAGON, #DD=SCORP, ..., #FF=отсутствует

HL=доступные видеорежимы (накладываются по OR):

 ;1=512x192 BW VMG

 ;2=384x304 ZX AC

 ;4=256x192 15 AC

 ;8=320x200 16 ATM

Остальные значения и биты пока зарезервированы.

                        КОHЕЦ СТАHДАРТА

Прилагаемый загрузчик пока не очень устойчиво работает,

принимаются советы по повышению устойчивости его работы

(впоследствии будем прошивать его в ПЗУ). Ограничения этого

загрузчика:

- не более 30 сессий на диске (информация берётся из последней);

- не более #8000 байт расстояние от начала корневого каталога до

конца описателя AUTORUN.ZX;

- IY, IM1 и #7ffd предполагаются уже установленными.

;загрузчик файла "AUTORUN.ZX" с CD по адресу #6000

device=#B0

comptype=1

idetype=1

language=1

COVPORT=#FB

videomodes=8

;===============ниже поддержаны только NEMO,SMUC,ATM============

        IF0 comptype-1

atm=1

        ELSE

atm=0

        ENDIF

        IF0 idetype-2

smuc=1|atm

        ELSE

smuc=atm

        ENDIF

       IFN smuc

      IFN atm

;схема ATM:

hddstat=#FEEF

hddcmd=#FEEF

hddhead=#FECF

hddcylhi=#FEAF

hddcyllo=#FE8F

hddsec=#FE6F

hddcount=#FE4F

hdderr=#FE2F

hdddatlo=#FE0F

hdddathi=#FF0F

 hddupr=#FEBE ;при установленном b7 FFBA

 hdduprON=#FFBA

 hddupr1=#F7

 hddupr0=#77

      ELSE

;схема SMUC:

hddstat=#FFBE

hddcmd=#FFBE

hddhead=#FEBE

hddcylhi=#FDBE

hddcyllo=#FCBE

hddsec=#FBBE

hddcount=#FABE

hdderr=#F9BE

hdddatlo=#F8BE

hdddathi=#D8BE

 hddupr=#FEBE ;при установленном b7 FFBA

 hdduprON=#FFBA

 hddupr1=#F7

 hddupr0=#77

      ENDIF

       ELSE

;схема Nemo:

hddstat=#F0

hddcmd=#F0

hddhead=#D0

hddcylhi=#B0

hddcyllo=#90

hddsec=#70

hddcount=#50

hdderr=#30

hdddatlo=#10

hdddathi=#11

 hddupr=#C8

 hdduprON=0

       ENDIF

SECBUF=#6000

        ORG #E800

begin

GO

        LD SP,#6000

        LD (IY+1),#CC

        EI

        HALT

        XOR A

        OUT (-2),A

        LD DE,#5801

        LD H,D,L,A

        LD BC,767

        LD (HL),L

        LDIR

       IFN atm

        CALL #3D46   ;вызов "проверочной" точки в (v)TR-DOS

        OR A         ;если не 0,vTR-DOS нет и ПЗУ не подменяем

        CALL Z,#3C98 ;вызов триггера подмены ПЗУ TRD/vTRD

       ENDIF

;инициализация

        LD A,device ;#B0=slave

        LD BC,hddhead

        CALL OUT_A

        CALL NO_BSY

    ;LD A,#FE

    ;IN A,(-2)

    ;RRA

    ;JNC SKIPINI

      ;LD A,#10        ;

      ;CALL HDSC       ;ZET9 для HDD

        LD A,#08        ;

        CALL HDSC       ;SMT для CD (сброс для ATAPI)

      ;RET C ;ATAPI only

        LD A,device ;#B0=slave   ;

        LD BC,hddhead            ;BUDDER

        CALL OUT_A               ;иначе при нераскрученном CD

                                 ;не выдаст EB14!!!

        LD A,#EC ;identify

        CALL HDSC

      ;RET C ;для CD выдаст ошибку (VEGA)

        LD B,30

        HALT

        DJNZ $-1

        CALL LEN_TO_HL

        LD BC,#EB14

        OR A

        SBC HL,BC

       JP NZ,0 ;HDD

       ;LD A,1

       ;OUT (-2),A

        LD HL,AP_1x

        CALL SEND_ATAPI

        LD B,30

        HALT

        DJNZ $-1

SKIPINI

        DI

       ;LD A,2

       ;OUT (-2),A

      ;сейчас номер сектора=0,0,0,0

        CALL READCDSECBUF ;иначе не работает READTOC

        LD HL,AP_READTOC

        CALL SEND_ATAPI

      ;RET C

        CALL NO_BSY

      ;RET C

        CALL WAIT_DRQ

      ;RET C

        CALL LEN_TO_HL

        LD B,H,C,L

        LD HL,SECBUF

        CALL TRANS_IN

        CALL NO_BSY

       ;LD A,3

       ;OUT (-2),A

     ;берем посл сессию (предпосл запись)

        LD HL,SECBUF+1

        LD A,(HL)

        ADD A,-10

        LD L,A

        LD D,#87

       ;CY=1

        CALL LOADER ;грузим начало сессии (#8800 байт)

        LD BC,#80A2

        ADD HL,BC

       ;CY=0

        CALL LOADER ;грузим корневой каталог

       ;HL=SECBUF

       ;LD A,4

       ;OUT (-2),A

FNDIDLOOP

       PUSH HL

        LD C,33

        ADD HL,BC

        LD DE,autorunname

        LD C,autorunnamesz

FNDID0  LD A,(DE)

        CPI

       JNZ FNDIDN

        INC DE

        JP PE,FNDID0

       POP HL

        LD C,6

        ADD HL,BC

       ;CY=0

       ;LD A,5

       ;OUT (-2),A

        CALL LOADER ;грузим autorun.zx

       PUSH HL

        LD A,device ;#B0=slave

        LD BC,comptype*256+idetype ;1=ATM,1=ATM

        LD DE,language*256+COVPORT ;1=RUS,#FB

        LD HL,videomodes ;8=ATM

       RET

FNDIDN

       POP HL

        LD C,(HL)

        INC HL

        LD B,(HL)

        DEC HL

       LD A,B

       OR C

       JNZ NOPADDING

        INC H

        LD L,A

NOPADDING

        ADD HL,BC

        LD A,H

        CP 64

       JNC FNDIDLOOP

        RST 0

LOADER

;грузим файл

;HL указывает на поле координат файла в motorola порядке

;CY=1: размер в DE

;CY=0: размер в (HL+4)

       PUSH DE

        LD DE,SECTOR

        LD BC,4

        LDIR

       POP DE

        JC $+5

         LD E,(HL)

         INC HL

         LD D,(HL)

        EXD

        SCF

        LD DE,2048

        INC B

        SBC HL,DE

        JNC $-3

        LD HL,SECBUF

       PUSH HL

LOADER0 PUSH BC

        CALL READCD

        PUSH HL

        LD HL,SECTOR+3

        INC (HL)

        DEC HL

        JZ $-2

        POP HL

        POP BC

        DJNZ LOADER0

       POP HL

        RET

;***************************************************************

OUT_A

       IFN smuc

      IFN atm

        PUSH AF,BC

        CALL OPENPORTS

        POP BC,AF

        OUT (C),A

        JR CLOSEPORTS

OPENPORTS

        LD A,%10101011

        LD BC,#2A53

        PUSH BC

        LD BC,#4177

        JP #3D2F

      ELSE

        LD IX,#3FF0

        PUSH IX

        JP #3D2F

      ENDIF

       ELSE ;nemo

        OUT (C),A

        RET

       ENDIF

IN_HDDSTAT

        LD BC,hddstat

IN_A

       IFN smuc

      IFN atm

        PUSH BC

        CALL OPENPORTS

        POP BC

        IN A,(C)

CLOSEPORTS

        PUSH AF,BC

        LD A,%10101011

        LD BC,#FF77

        OUT (C),A

        POP BC,AF

        RET

      ELSE

        LD IX,#3FF3

        PUSH IX

        JP #3D2F

      ENDIF

       ELSE

        IN A,(C)

        RET

       ENDIF

;ПОСЛАТЬ КОМАHДУ HА ВИHТ

HDSC

        LD BC,hddcmd

        CALL OUT_A

;ОЖИДАHИЕ ОСВОБОЖДЕHИЯ УСТРОЙСТВА

NO_BSY

        CALL IN_HDDSTAT

        RLCA

       RET NC

        JR NO_BSY

;ОЖИДАHИЕ ГОТОВHОСТИ ПЕРЕДАЧИ ДАHHЫХ

WAIT_DRQ

        CALL IN_HDDSTAT

        BIT 3,A

       RET NZ

        JR WAIT_DRQ

;ЧТЕHИE ЧИСЛА ИЗ РЕГИСТРА ЦИЛИHДРА

LEN_TO_HL

        LD BC,hddcyllo

        CALL IN_A

        LD L,A

        LD BC,hddcylhi

        CALL IN_A

        LD H,A

        RET

;IN:HL-АДРЕС ПРИЕМА ДАHHЫХ

;   BC-КОЛ-ВО БАЙТ

TRANS_IN

        PUSH BC

        CALL NO_BSY

       ;POP BC

      ;RET C

       ;PUSH BC

        CALL WAIT_DRQ

        POP BC

      ;RET C

        INC BC

        SRL B

        RR C

TR_IN0  PUSH BC

        LD BC,hdddatlo

        CALL IN_A

        LD (HL),A

        INC HL

        LD BC,hdddathi

        CALL IN_A

        LD (HL),A

        POP BC

        CPI

        JP PE,TR_IN0

        RET

;ПЕРЕДАЧА ATAPI-ПАКЕТА

SEND_ATAPI

      ;LD A,device ;#B0=slave

      ;LD BC,hddhead

      ;CALL OUT_A

        LD BC,hddcyllo

        XOR A

        CALL OUT_A

        LD BC,hddcylhi

        LD A,'2048

        CALL OUT_A

        LD A,#A0

        CALL HDSC

      ;RET C

        LD B,6

TR_OUT0 PUSH BC

        INC HL

        LD A,(HL)

        LD BC,hdddathi

        CALL OUT_A

        DEC HL

        LD A,(HL)

        LD BC,hdddatlo

        CALL OUT_A

        INC HL,HL

        POP BC

        DJNZ TR_OUT0

        RET

READCDSECBUF

        LD HL,SECBUF

READCD

;ЧТЕHИЕ СЕКТОРА

lOAD_SECTOR

       PUSH HL

        LD HL,AP_READ

        CALL SEND_ATAPI

       POP HL

      ;RET C

;бывает ситуация,что CHECK CONDITION (D0 статуса)=0,

;а при этом DRQ не выдается!

       ;CALL IN_HDDSTAT

       ;RRA

      ;RET C

        CALL NO_BSY ;иначе виснет при иниц-ции CD

      ;RET C

       ;CALL WAIT_DRQ ;здесь виснет, если инитить слишком рано

      ;RET C     ;после выхода разгоняется, ждет кнопку и читает

     ;ожидание DRQ по рецепту Budder'а

        LD DE,0

RDCDDRQ CALL IN_HDDSTAT

        BIT 3,A

       JNZ READ_P2

        INC DE

        BIT 2,D ;4,D

        JZ RDCDDRQ

        PUSH HL

        LD HL,AP_00

        CALL SEND_ATAPI

        POP HL

        JR lOAD_SECTOR

READ_P2

        LD BC,2048

        CALL TRANS_IN

        JP NO_BSY

autorunname

        DB "AUTORUN.ZX"

autorunnamesz=$-autorunname

;ATAPI-ПАКЕТ "ПУСТЫШКА"

AP_00

        DW 0

        DS 10

;ATAPI-ПАКЕТ "SPEED 1x"

AP_1x

        DW #BB

        DB 0

        DB 176 ;1x=176k/s

        DS 8

;ATAPI-ПАКЕТ "ЧТЕHИЕ"

AP_READ

       IFN 0

        DW #BE   ;"READ CD"

SECTOR  DB 0,0,0,0

        DB 0

        DB 0,1   ;=1 сектор

        DB #10   ;читаем только данные

        DB 0,0

       ELSE

        DW #28   ;"READ(10)"

SECTOR  DB 0,0,0,0

        DB 0

        DB 0,1   ;=1 сектор

        DB 0

        DB 0,0

       ENDIF

AP_READTOC

        DW #43   ;SCMSF=0,т.е.секторы,а не MSF

        DB 0     ;FORMAT=0:все сессии

        DS 3

        DB 0     ;с 1-й сессии

        DB '2048 ;длина табл

        DB 2048

        DB #00   ;FUNC

        DS 2

end

       DISPLAY end-begin

        ORG #5CDD

        DB "boot    B

        INCLUDE "mrip*",#C0

